Black Scholes PDE Derivation 12-31-10

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[3]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Classic derivation of the Black-Scholes PDE

The following derivation is sufficiently simple that *Mathematica* is not really warranted. It can be performed in a few minutes with pen and paper.

Nonetheless, it is worthwhile to illustrate the *Mathematica* techniques involved since this derivation of this PDE is the template other financially relevant partial differential equations for which considerably more symbolic manipulation is required.

Assume that S follows a drifted geometric Brownian motion stochastic process

```
sde[S] = dS / S == μ dt + σ dz
```

$$\frac{dS}{S} == dt\,\mu + dz\,\sigma$$

We seek to develop a partial differential equation for the contingent claim F(S, t). Start by developing a power series expansion for F.

```
w1[1] = F[S, t] == Series[F[S, t], {S, S0, 2}, {t, t0, 1}] // Normal
```

$$F[S,t] == F[S0,t0] + (t-t0)\,F^{(0,1)}[S0,t0] + (S-S0)\left(F^{(1,0)}[S0,t0] + (t-t0)\,F^{(1,1)}[S0,t0]\right) + (S-S0)^2\left(\frac{1}{2}F^{(2,0)}[S0,t0] + \frac{1}{2}(t-t0)\,F^{(2,1)}[S0,t0]\right)$$

Introduce some simplifying variables. The last replacement is made to convert the equation into an expression. Then operations can be performed on a single expression, as opposed to performing identical operations on the expressions on both sides an equation.

```
w1[2] =
 w1[1] /. {F[S, t] → dF + F[S0, t0], S → dS + S0, t → dt + t0} /. Equal → Subtract
```

$$dF - dt\,F^{(0,1)}[S0,t0] - dS\left(F^{(1,0)}[S0,t0] + dt\,F^{(1,1)}[S0,t0]\right) - dS^2\left(\frac{1}{2}F^{(2,0)}[S0,t0] + \frac{1}{2}dt\,F^{(2,1)}[S0,t0]\right)$$

```
w1[3] = w1[2] /. Solve[sde[S], dS][[1, 1]]
```

$$dF - dt\,F^{(0,1)}[S0,t0] - S\,(dt\,\mu + dz\,\sigma)\left(F^{(1,0)}[S0,t0] + dt\,F^{(1,1)}[S0,t0]\right) - S^2\,(dt\,\mu + dz\,\sigma)^2\left(\frac{1}{2}F^{(2,0)}[S0,t0] + \frac{1}{2}dt\,F^{(2,1)}[S0,t0]\right)$$

Introduce the diffusive ordering consistent with Ito. Here $\epsilon << 1$

```
w1[4] = w1[3] /. dt -> ϵ dt /. dz → ϵ^(1/2) dz /. S0 → S /. t0 → t
```

$$dF - dt\,\epsilon\,F^{(0,1)}[S,t] - S\left(dt\,\epsilon\,\mu + dz\sqrt{\epsilon}\,\sigma\right)\left(F^{(1,0)}[S,t] + dt\,\epsilon\,F^{(1,1)}[S,t]\right) - S^2\left(dt\,\epsilon\,\mu + dz\sqrt{\epsilon}\,\sigma\right)^2\left(\frac{1}{2}F^{(2,0)}[S,t] + \frac{1}{2}dt\,\epsilon\,F^{(2,1)}[S,t]\right)$$

Introduce the Ito ordering and truncate the expansion

```
w1[5] = ExpandAll[w1[4]] /. ϵ^n_/;n>1 -> 0 /. ϵ -> 1 /. dz^2 → dt
```

$$dF - dt\,F^{(0,1)}[S,t] - dt\,S\,\mu\,F^{(1,0)}[S,t] - dz\,S\,\sigma\,F^{(1,0)}[S,t] - \frac{1}{2}dt\,S^2\,\sigma^2\,F^{(2,0)}[S,t]$$

Form a hedge portfolio by adding an amount Δ of the underlier P = F + Δ S. The rate of change of this portfolio under the same ordering is

```
w1[6] = dP == dF + Δ dS /. Solve[sde[S], dS][[1, 1]] /. dt -> ϵ dt /. dz → ϵ^(1/2) dz /.
    ϵ^n_/;n>1 -> 0 /. ϵ -> 1
```

$$dP == dF + S\,\Delta\,(dt\,\mu + dz\,\sigma)$$

In detail

```
w1[7] = w1[6] /. Solve[w1[5] == 0, dF][[1, 1]]
```

$$dP == S\,\Delta\,(dt\,\mu + dz\,\sigma) + \frac{1}{2}\left(2\,dt\,F^{(0,1)}[S,t] + 2\,dt\,S\,\mu\,F^{(1,0)}[S,t] + 2\,dz\,S\,\sigma\,F^{(1,0)}[S,t] + dt\,S^2\,\sigma^2\,F^{(2,0)}[S,t]\right)$$

or

```
w1[8] = w1[7][[1]] == Collect[ExpandAll[w1[7][[2]]], {dt, dz}]
```

$$dP == dz\,\left(S\,\Delta\,\sigma + S\,\sigma\,F^{(1,0)}[S,t]\right) + dt\left(S\,\Delta\,\mu + F^{(0,1)}[S,t] + S\,\mu\,F^{(1,0)}[S,t] + \frac{1}{2}S^2\,\sigma^2\,F^{(2,0)}[S,t]\right)$$

We see that the stochastic risk of this portfolio can be removed by choosing α such that the term multiplying dz vanishes

```
w1[9] = Solve[Coefficient[w1[8][[2]], dz] == 0, Δ] [[1, 1]]
```

$$\Delta \to -F^{(1,0)}[S,t]$$

Then the hedge portfolio is

```
w1[10] = w1[8] /. w1[9]
```

$$dP == dt\left(F^{(0,1)}[S,t] + \frac{1}{2}S^2\,\sigma^2\,F^{(2,0)}[S,t]\right)$$

Notice the remarkable feature that this expression does not depend on the drift μ. The assumption is then made that the portfolio should grow according to the risk free interest rate. The contingent claim F grows at rate r, but the stock hedge must be adjusted for the dividend yield.

```
w1[11] = dP == (r F[S, t] + (r - q) Δ S ) dt
```

$$dP == dt\,\left((-q+r)\,S\,\Delta + r\,F[S,t]\right)$$

or

```
w1[12] = w1[11] /. w1[9]
```

$$dP == dt\,\left(r\,F[S,t] - (-q+r)\,S\,F^{(1,0)}[S,t]\right)$$

Combining these two expressions for dP

```
w1[13] = w1[10] /. (w1[12] /. Equal → Rule)
```

$$dt\,\left(r\,F[S,t] - (-q+r)\,S\,F^{(1,0)}[S,t]\right) == dt\left(F^{(0,1)}[S,t] + \frac{1}{2}S^2\,\sigma^2\,F^{(2,0)}[S,t]\right)$$

or

```
w1[14] = w1[13] /. Equal → Subtract /. dt → 1
```

$$r\,F[S,t]-F^{(0,1)}[S,t]-(-q+r)\,S\,F^{(1,0)}[S,t]-\frac{1}{2}S^2\,\sigma^2\,F^{(2,0)}[S,t]$$

Let's reverse the sign and form an equation.

```
w1[15] = -w1[14] == 0
```

$$-r\,F[S,t]+F^{(0,1)}[S,t]+(-q+r)\,S\,F^{(1,0)}[S,t]+\frac{1}{2}S^2\,\sigma^2\,F^{(2,0)}[S,t]==0$$

Introduce a subscript notation for the partial derivatives

```
w1[16] =
 w1[15] /. {F^(0,1)[S, t] → F_t[S, t], F^(1,0)[S, t] → F_S[S, t], F^(2,0)[S, t] → F_SS[S, t]}
```

$$-r\,F[S,t]+(-q+r)\,S\,F_S[S,t]+\frac{1}{2}S^2\,\sigma^2\,F_{SS}[S,t]+F_t[S,t]==0$$

and we have the classical Black Scholes PDE. This can be display in more standard mathematical notation

```
w1[16] // PhysicsForm
```

$$S\,(r-q)\,F_S(S,t)-r\,F(S,t)+\frac{1}{2}\sigma^2\,S^2\,F_{SS}(S,t)+F_t(S,t)=0$$